version 13
#delimit;
clear all;
set more off;
capture log close;


global strformat "%8.3f";

capture program drop sig_p;
program sig_p;
   args obj_var point_est_var point_sd_var p_val row_id_var row_num;
   local point_est `point_est_var';
   local point_sd `point_sd_var';
   if `p_val' > 0.1 {;
      replace `obj_var' = string(`point_est', "$strformat") if `row_id_var' == `row_num';
      };
   if `p_val' > 0.05 & `p_val' <= 0.1  {;
      replace `obj_var' = string(`point_est', "$strformat")+"*" if `row_id_var' == `row_num';
      };
   if `p_val' > 0.01 & `p_val' <= 0.05  {;
      replace `obj_var' = string(`point_est', "$strformat")+"**" if `row_id_var' == `row_num';
      };
   if `p_val' <= 0.01 {;
      replace `obj_var' = string(`point_est', "$strformat")+"***" if `row_id_var' == `row_num';
      };
   replace `obj_var' = "<" + string(`point_sd', "$strformat") + ">" + " " if `row_id_var' == `row_num' + 1;
   end;


*enter directory name here;

use "$dir/ReStat_children.dta", clear;


global fu_child_controls_cm
	"control_fu_child_age control_fu_female
	missing_fu_child_age missing_fu_female";
		
global fu_hh_controls_cm
	"control_fu_adults control_fu_hh_head_edu control_fu_hh_head_occ_farmer 
	control_fu_total_land control_fu_household_size
	missing_fu_adults missing_fu_hh_head_edu missing_fu_hh_head_occ_farmer 
	missing_fu_total_land missing_fu_household_size";



generate row_num = _n;



forvalues i=0/25 {;
	generate col`i' = "";
};

local row = 4;
sort row_num;
order col0-col25;



foreach var in

	bl_child_age bl_child_female bl_child_in_school {; 
	
	replace col0 = "`var'" if row_num == `row';
	
 	su `var' if bl_child_level == 1 & pooled_treatment == 0 & baseline_household == 1;
	replace col2 = string(r(mean), "$strformat") if row_num == `row';

	regress `var' pooled_treatment if bl_child_level == 1 & baseline_household == 1, cluster(villcode); 
	test pooled_treatment;
	sig_p col3 _b[pooled_treatment] _se[pooled_treatment] r(p) row_num `row';
	
	local row = `row' + 2;
	
};

local row = `row' + 3;

foreach var in 

	bl_household_size bl_number_child bl_education bl_farmer {; 
	
	replace col0 = "`var'" if row_num == `row';
	
 	su `var' if bl_child_level == 1 & pooled_treatment == 0 & baseline_household == 1;
	replace col2 = string(r(mean), "$strformat") if row_num == `row';

	regress `var' pooled_treatment if bl_child_level == 1 & baseline_household == 1, cluster(villcode); 
	test pooled_treatment;
	sig_p col3 _b[pooled_treatment] _se[pooled_treatment] r(p) row_num `row';
	
	local row = `row' + 2;
	
};




local row = 4;


foreach var in 
	fu_child_age fu_female bl_child_in_school fu_child_of_hh_head {; 
	
	replace col0 = "`var'" if row_num == `row';
	
 	su `var' if fu_young_child == 1 & fu_child_level == 1 & pooled_treatment == 0 [aw=hh_weight];
	replace col5 = string(r(mean), "$strformat") if row_num == `row';

	regress `var' pooled_treatment if fu_young_child == 1 & fu_child_level == 1 [pw=hh_weight], cluster(villcode); 
	test pooled_treatment;
	sig_p col6 _b[pooled_treatment] _se[pooled_treatment] r(p) row_num `row';
	
	local row = `row' + 2;
	
};

local row = `row' + 1;

foreach var in 
	
	fu_household_size fu_num_children fu_hh_head_edu fu_hh_head_occ_farmer 
	fu_total_land {; 
	
	replace col0 = "`var'" if row_num == `row';
	
 	su `var' if fu_young_child == 1 & fu_child_level == 1 & pooled_treatment == 0 [aw=hh_weight];
	replace col5 = string(r(mean), "$strformat") if row_num == `row';

	regress `var' pooled_treatment if fu_young_child == 1 & fu_child_level == 1 [pw=hh_weight], cluster(villcode); 
	test pooled_treatment;
	sig_p col6 _b[pooled_treatment] _se[pooled_treatment] r(p) row_num `row';
	
	local row = `row' + 2;
	
};

local row = `row' + 1;

foreach var in 
	
	fu1c_ht_pukka fu1c_ht_semi_pukka fu1c_ht_kaccha fu1c_ht_thatched_huts 
	fu_num_goats fu_muslim_fiqa_sunni fu_lang_urdu fu_lang_sindhi {; 
	
	replace col0 = "`var'" if row_num == `row';
	
 	su `var' if fu_young_child == 1 & fu_child_level == 1 & pooled_treatment == 0 [aw=hh_weight];
	replace col5 = string(r(mean), "$strformat") if row_num == `row';

	regress `var' pooled_treatment if fu_young_child == 1 & fu_child_level == 1 [pw=hh_weight], cluster(villcode); 
	test pooled_treatment;
	sig_p col6 _b[pooled_treatment] _se[pooled_treatment] r(p) row_num `row';
	
	local row = `row' + 2;
	
};

local row = `row' + 1;

reg pooled_treatment 
	bl_child_age bl_child_female bl_child_in_school
	bl_household_size bl_number_child bl_education bl_farmer
	if bl_child_level == 1 & baseline_household == 1, cluster(villcode); 
test
	bl_child_age = bl_child_female = bl_child_in_school =
	bl_household_size = bl_number_child = bl_education = bl_farmer = 0;
replace col3 = string(r(F), "$strformat") if row_num == `row';
replace col3 = string(r(p), "$strformat") if row_num == `row' + 1;



reg pooled_treatment 
	fu_child_age fu_female bl_child_in_school fu_child_of_hh_head
	fu_household_size fu_num_children fu_hh_head_edu fu_hh_head_occ_farmer 
	fu_total_land
	fu1c_ht_pukka fu1c_ht_semi_pukka fu1c_ht_kaccha fu1c_ht_thatched_huts 
	fu_num_goats fu_muslim_fiqa_sunni fu_lang_urdu fu_lang_sindhi
	if fu_young_child == 1 & fu_child_level == 1 [pw=hh_weight], cluster(villcode); 
test
	fu_child_age = fu_female = bl_child_in_school = fu_child_of_hh_head =
	fu_household_size = fu_num_children = fu_hh_head_edu = fu_hh_head_occ_farmer =
	fu_total_land =
	fu1c_ht_pukka = fu1c_ht_semi_pukka = fu1c_ht_kaccha = fu1c_ht_thatched_huts = 
	fu_num_goats = fu_muslim_fiqa_sunni = fu_lang_urdu = fu_lang_sindhi = 0;
replace col6 = string(r(F), "$strformat") if row_num == `row';
replace col6 = string(r(p), "$strformat") if row_num == `row' + 1;

local row = `row' + 2;

outsheet col0-col6 using "$dir/Table 2 Treat Balance.csv" if
	row_num < `row', comma replace;
	
